CREATE OR REPLACE PACKAGE msg
/*********************************************************************
* Author:   Bill Coulam
* Revised:  2002Aug23
*
* Description:
*   A collection of routines meant to replace the awful 
* dbms_output.put_line built-in.
*********************************************************************/
IS
 
-- Public constants, types, cursors, procedures, functions, etc.
DATETIMEMASK CONSTANT VARCHAR2(18) := 'YYYYMonDD HH24:MI:SS';
NUMBERMASK   CONSTANT VARCHAR2(25) := 'FM999G999G999G990D0009';

----------------------------------------------------------------------
-- p
--    The first incarnation of the [p]rint procedure, the one that only
-- takes 1 or 2 strings, was designed to be able to handle character strings
-- longer than 255 chars, but less than 32,767 chars, which dbms_output's
-- put_line routine cannot do.
--
-- The remaining versions of [p]rint were designed
-- (1) for type-overridden name consistency so a whole API didn't need to be learned,
-- (2) to be a whole lot shorter to type than "dbms_output.put_line" and
-- (3) to shorten some common uses for dbms_output.
-- NOTE! The other versions of [p]rint that have a VARCHAR2 parameter were NOT
-- designed to handle large VARCHAR2 values.
--
-- For example, most of them are designed to print out a little string tag,
-- followed by the value. This is very useful for manual debugging, showing
-- the name of a variable, then showing the variable's value. The p routine
-- eliminates a little extra typing by taking care of the concatenations
-- for you.
-- Example:
--    msg.p('id_my_date',id_my_date);
--                VS.
--    dbms_output.put_line('id_my_date'||' : '||id_my_date);
----------------------------------------------------------------------
PROCEDURE p
(
   is_str   IN VARCHAR2 --  will be printed to stdout
   ,is_str2 IN VARCHAR2 DEFAULT NULL
);
PROCEDURE p
(
   id_date  IN DATE --  will be printed to stdout, using format
   ,is_fmt  IN VARCHAR2 DEFAULT DATETIMEMASK

);
PROCEDURE p
(
   in_num   IN NUMBER --  will be printed to stdout, using format
   ,is_fmt  IN VARCHAR2 DEFAULT NUMBERMASK
);
PROCEDURE p
(
   ib_bool  IN BOOLEAN --  will be converted to "TRUE", "FALSE" or NULL
);
PROCEDURE p
(
   is_str   IN VARCHAR2 --  used as label, preceding date value
   ,id_date  IN DATE
   ,is_fmt  IN VARCHAR2 DEFAULT DATETIMEMASK
);
PROCEDURE p
(
   is_str   IN VARCHAR2 --  used as label, preceding numeric value
   ,in_num   IN NUMBER
   ,is_fmt  IN VARCHAR2 DEFAULT NUMBERMASK
);
PROCEDURE p
(
   is_str   IN VARCHAR2 --  used as label, preceding boolean (string) value
   ,ib_bool  IN BOOLEAN --  will be converted to "TRUE", "FALSE" or NULL
);
----------------------------------------------------------------------
-- set_tvsepchar (Set Tag Value Seperator Character)
--    Allows one to change the character(s) used to separate tags from
-- values in the four versions of "p" that concatenate a string to
-- a variable value. The default sepchar is a colon.
--
-- NOTE! The internal variable used as the sepchar is currently limited
-- to 30 characters at a maximum.
----------------------------------------------------------------------
PROCEDURE set_tvsepchar
(
   is_sepchar  IN VARCHAR2 -- defaults to ': '
);
----------------------------------------------------------------------
-- set_linelen
--    Allows one to change the length of individual lines spat out to
-- stdout by the "p" routines. This defaults to 80.
----------------------------------------------------------------------
PROCEDURE set_linelen
(
   in_linelen  IN NUMBER -- defaults to 80
);

END msg;
/


CREATE OR REPLACE PACKAGE BODY msg
/*********************************************************************
* Author:   Bill Coulam
* Revised:  2002Aug23
*********************************************************************/
IS 

/* package global variables, cursors, etc. -------------------------*/
gn_linelen  NUMBER(3,0) := 80;
gs_tvsepchar VARCHAR2(30) := ': ';

/* private procedures and functions --------------------------------*/
FUNCTION fs_btostr(ib_bool_val IN BOOLEAN) RETURN VARCHAR2
IS
BEGIN
   IF (ib_bool_val) THEN
      RETURN 'TRUE';
   ELSIF (ib_bool_val IS NULL) THEN
      RETURN 'NULL';
   ELSE
      RETURN 'FALSE';
   END IF;
END fs_btostr;

/* public procedures and functions ---------------------------------*/
----------------------------------------------------------------------
PROCEDURE p
(
   is_str   IN VARCHAR2 --  will be printed to stdout
   ,is_str2 IN VARCHAR2 DEFAULT NULL
)
IS
   MAX_LINE CONSTANT NUMBER := gn_linelen;  -- maximum #chars put_line can handle
   start_text NUMBER;  -- starting pt of text to print
   end_text   NUMBER;  -- ending pt of text to print
   lentxt     NUMBER;  -- length from logical start to logical end of text
   break_pt   NUMBER;  -- break pt (found whitespace)
   start_next NUMBER;  -- next non-newline char


   CR CONSTANT VARCHAR2(1) := CHR(13);  -- carriage return character
   LF CONSTANT VARCHAR2(1) := CHR(10);  -- linefeed character
   SP CONSTANT VARCHAR2(1) := CHR(32);  -- space character
   DS CONSTANT VARCHAR2(2) := '  ';     -- 2 spaces
   TB CONSTANT VARCHAR2(1) := CHR(9);   -- tab character
BEGIN
   IF (is_str2 IS NOT NULL) THEN
      p(is_str||gs_tvsepchar||is_str2);
   ELSE

      start_text := 1;
      LOOP
         end_text := start_text + MAX_LINE - 1;
         lentxt := NVL(LENGTH(SUBSTR(is_str, start_text, MAX_LINE)),0);

         IF (lentxt < MAX_LINE) THEN  -- last chunk of text in string
            DBMS_OUTPUT.put_line(SUBSTR(is_str, start_text, lentxt));
            EXIT;  -- and we're done!
         ELSE  -- not done yet so find good break pt
            break_pt := 0;  -- reset
            FOR i IN REVERSE start_text .. end_text LOOP
               IF (SUBSTR(is_str, i, 1) IN (CR, LF, SP, TB)) THEN
                  break_pt := i;  -- found suitable break pt
                  EXIT;
               END IF;
            END LOOP;  -- find break pt

            IF (break_pt = 0) THEN  -- no suitable break pt found!
               DBMS_OUTPUT.put_line(SUBSTR(is_str, start_text, MAX_LINE));
               start_text := end_text + 1;  -- next start pt
            ELSE  -- print to just before break pt
               DBMS_OUTPUT.put_line(SUBSTR(is_str, start_text, break_pt-start_text));
               start_next := 0;  -- reset

               FOR i IN break_pt+1 .. end_text LOOP  -- find next non-newline char
                  IF (SUBSTR(is_str, i, 1) NOT IN (CR, LF)) THEN
                    start_next := i;
                    EXIT;
                  END IF;
               END LOOP;  -- find next non-newline char

               IF (start_next = 0) THEN  -- no non-newline char found
                  start_text := end_text + 1;
               ELSE
                  start_text := start_next;  -- start at non-newline char found
               END IF;
            END IF;  -- break pt?
         END IF;  -- last chunk?
      END LOOP;  -- print long string
   END IF; -- print tag/value pair, or long string?
EXCEPTION
WHEN OTHERS THEN
   BEGIN
      DBMS_OUTPUT.put_line('msg.p ERROR: ' || SQLERRM(SQLCODE));
   EXCEPTION
   WHEN OTHERS THEN
      NULL;  -- don't care
   END;
END p;

----------------------------------------------------------------------
PROCEDURE p
(
   id_date  IN DATE --  will be printed to stdout, using format
   ,is_fmt  IN VARCHAR2 DEFAULT DATETIMEMASK

)
IS
BEGIN
   p(TO_CHAR(id_date,is_fmt));
END p;

----------------------------------------------------------------------
PROCEDURE p
(
   in_num   IN NUMBER --  will be printed to stdout, using format
   ,is_fmt  IN VARCHAR2 DEFAULT NUMBERMASK
)
IS
BEGIN
   p(TO_CHAR(in_num,is_fmt));
END p;

----------------------------------------------------------------------
PROCEDURE p
(
   ib_bool  IN BOOLEAN --  will be converted to "TRUE", "FALSE" or NULL
)
IS
BEGIN
   p(fs_btostr(ib_bool));
END p;

----------------------------------------------------------------------
PROCEDURE p
(
   is_str   IN VARCHAR2 --  used as label, preceding date value
   ,id_date  IN DATE
   ,is_fmt  IN VARCHAR2 DEFAULT DATETIMEMASK
)
IS
BEGIN
   p(is_str||gs_tvsepchar||TO_CHAR(id_date,is_fmt));
END p;

----------------------------------------------------------------------
PROCEDURE p
(
   is_str   IN VARCHAR2 --  used as label, preceding numeric value
   ,in_num   IN NUMBER
   ,is_fmt  IN VARCHAR2 DEFAULT NUMBERMASK
)
IS
BEGIN
   p(is_str||gs_tvsepchar||TO_CHAR(in_num,is_fmt));
END p;

----------------------------------------------------------------------
PROCEDURE p
(
   is_str   IN VARCHAR2 --  used as label, preceding boolean (string) value
   ,ib_bool  IN BOOLEAN --  will be converted to "TRUE", "FALSE" or NULL
)
IS
BEGIN
   p(is_str||gs_tvsepchar||fs_btostr(ib_bool));
END p;

----------------------------------------------------------------------
PROCEDURE set_tvsepchar
(
   is_sepchar  IN VARCHAR2
)
IS
BEGIN
   gs_tvsepchar := NVL(is_sepchar,gs_tvsepchar);
END set_tvsepchar;

----------------------------------------------------------------------
PROCEDURE set_linelen
(
   in_linelen  IN NUMBER
)
IS
BEGIN
   gn_linelen := NVL(in_linelen,gn_linelen);
END set_linelen;

END msg;
/
